

SCIENCE & VIE

MENSUEL - N° 876

SEPTEMBRE 1990

L'ÉTÉ FOU
ET LA BIZARRE
COLÈRE
DU SOLEIL

● Les Français sont
les plus gros
consommateurs
d'antibiotiques

● L'ordinateur
(faux)tographe



Jégu.

● Les supersatellites,
vrais héros du
conflit du Golfe

M 2578 - 876 - 20,00 F



145, FB - 6 FS - \$Can 3.75 - 550 Ptas - 550 Esc - 23 Dh - 2.750 Dt - 4800 L - USA NYC \$ 3.95 - RCI : 1370 CFA - 940 CFP

Un jeu de quilles

INFORMATIQUE AMUSANTE

Nous vous proposerons ce mois-ci un petit programme de jeu. Inspiré du classique jeu de "casse-briques", il s'agira ici d'éliminer le plus grand nombre de quilles possibles à l'aide d'une balle rebondissant sur les bords du terrain. Pour cela, vous disposerez d'un crédit de trois balles. A vous d'en tirer profit en utilisant les flèches de déplacement horizontal du curseur pour piloter la raquette. Notons que si vous disposez d'une manette de jeu, il sera facile de modifier le programme de manière que cette dernière commande directement la raquette.

Passons maintenant à l'écriture

de ce programme. D'abord, nous redéfinirons des symboles graphiques afin de représenter les quilles, la raquette et les bords du terrain. Ensuite, nous afficherons à l'écran les contours de ce dernier grâce à deux boucles FOR-NEXT. La première tracera ses limites horizontales, la seconde les verticales. De même, toujours à l'aide d'une boucle FOR-NEXT, nous disposerons les quilles. Afin de mémoriser la position des quilles touchées au cours de la partie, nous créerons un tableau. Celui-ci comporte autant de cases que de quilles et sera, en fait, l'image de l'état du terrain. Si l'une de ses cases comporte un 0,

cela indique à l'ordinateur que la quille correspondante est présente ; si, en revanche, elle comporte un 1, la machine saura que la quille a été touchée.

Enfin, pour cette première phase d'initialisation du jeu, les positions horizontales de la raquette et de la balle seront fixées. Pour cette dernière, le choix s'effectuera de manière aléatoire afin d'obtenir un "engagement" différent pour chaque nouvelle balle.

A partir de là, le jeu peut débiter. En premier lieu, nous nous intéresserons à l'animation de la raquette. Sa position horizontale sera, en effet, commandée directement par incrément, ou décrétement, de la variable XR en fonction de l'état, enfoncé ou non, des touches 1 et 8, correspondant respectivement à "flèche vers la droite" et "flèche vers la gauche".

Vient à présent l'animation du

```

10 MODE 1: CLEAR
20 REM *****
30 REM *
40 REM * REDEFINITION DES *
50 REM * SYMBOLES GRAPHIQUES. *
60 REM *
70 REM *****
80 SYMBOL AFTER 129
90 SYMBOL 130,0,3,1,6,13,5,2,1
100 SYMBOL 131,0,192,128,96,176,160,64,128
110 SYMBOL 132,0,3,1,4,8,4,2,1
120 SYMBOL 133,0,192,128,32,16,32,64,128
130 SYMBOL 134,170,85,170,85,170,85,170,85
140 SYMBOL 135,7,31,127,254,254,254,96,63
150 SYMBOL 136,224,248,254,127,127,127,6,252
160 SYMBOL 137,60,102,195,129,129,195,102,60
170 REM *****
180 REM *
190 REM * DEFINITION DES *
200 REM * ENVELOPPES SONORES. *
210 REM *
220 REM *****
230 ENV 1,5,3,1,1,0,1,1,-15,1
240 ENV 2,1,15,1,3,-1,1,1,-12,1
250 LET SND=0
260 PRINT "ILLUSTRATION SONORE ? (O/N)
270 LET K$=UPPER$(INKEY$)
280 IF K$="" THEN GOTO 270
290 IF K$="O" THEN LET SND=1
300 CLS
310 REM *****
320 REM *
330 REM * INITIALISATION DES *
340 REM * VARIABLES. *
350 REM *
360 REM *****
370 LET CR=3:LET SC=0
380 LOCATE 25,1:PRINT "VOTRE SCORE:"
390 LOCATE 25,5:PRINT "CREDIT:";CR
400 REM *****
410 REM *
420 REM * TRACE DU TERRAIN *
430 REM * DE JEU. *
440 REM *
450 REM *****
460 FOR I=1 TO 22
470 LOCATE I,1:PRINT CHR$(134)
480 LOCATE I,25:PRINT CHR$(134)
490 NEXT I
500 REM *****
510 REM *
520 REM * BOUCLES DE POSITIONNEMENT *
530 REM * DES QUILLES. *
540 REM *
550 REM *****
560 FOR I=1 TO 25
570 LOCATE I,1:PRINT CHR$(134)
580 LOCATE 22,I:PRINT CHR$(134)
590 NEXT I
600 FOR I=2 TO 20 STEP 2
610 LOCATE I,5:PRINT CHR$(130);CHR$(131)
620 LOCATE I,6:PRINT CHR$(132);CHR$(133)
630 LOCATE I,7:PRINT CHR$(130);CHR$(131)
640 LOCATE I,8:PRINT CHR$(132);CHR$(133)
650 NEXT I
660 REM *****
670 REM *
680 REM * CREATION DU TABLEAU DE *
690 REM * MEMORISATION DE LA POSITION *
700 REM * DES QUILLES. *
710 REM *
720 REM *****
730 DIM B(10,4)
740 REM *****
750 REM *
760 REM * DEBUT DU JEU. *
770 REM *
780 REM *****
790 LET XR=11
800 LET XB=INT(RND*11)+5:LET YB=20
810 MXB=1-((INT(RND*2))*2):LET MYB=-1

```

mouvement de la balle. Deux variables XB, pour l'horizontale, et YB, pour la verticale, détermineront sa position instantanée. De même, MXB et MYB détermineront le sens des déplacements. C'est donc autour de ces deux dernières variables que tournera l'animation. En fait, elles ne pourront prendre que deux valeurs, à savoir +1 ou -1. Leur changement d'état s'effectuera lors de l'une des trois conditions suivantes : la balle touche le bord du terrain, la balle renverse une quille, ou, enfin, la balle est renvoyée par la raquette. Notons d'ailleurs que dès que la balle touche une quille, ce fait est immédiatement consigné dans notre tableau de mémorisation ; la case concernée passe immédiatement de 0 à 1. Ainsi, par la suite, si la balle revient sur cet emplacement elle continuera effectivement sa trajectoire au lieu de rebondir. De même,

le score sera "incrémenté" du nombre de points correspondants à la valeur de la quille, à savoir 5 pour celles du premier rang, puis 10 pour le deuxième, 20 pour le troisième et 25 pour le dernier.

Tant que le crédit de balles ne sera pas épuisé, notre jeu rebouclera. Sinon, celui-ci s'arrêtera et proposera une nouvelle partie.

La frappe de ce programme ne doit pas poser de problème particulier. Il est également relativement simple de l'adapter à d'autres machines que l'Amstrad 464, pour lequel il a été spécialement étudié. Rappelons simplement que MODE indique à l'ordinateur le nombre de caractères par lignes — dans notre cas, 40 —, SYMBOL permet de mémoriser un nouveau jeu de caractères graphiques en redéfinissant point par point chaque symbole, LOCATE est utilisé pour adresser le curseur et INKEY (x) permet de

savoir si une touche est enfoncée ou non. Si INKEY(x) vaut 0, cela indique que la touche numéro x est enfoncée, en cas contraire INKEY(x) vaudra -1. Hormis ces instructions un peu particulières ne concernant que l'Amstrad, le reste de basic utilisé est des plus communs et ne doit donc pas poser de problème d'adaptation.

L'utilisation de ce programme est également très simple. Après l'avoir entièrement tapé et demandé RUN, l'écran de jeu s'affichera et la balle sera lancée. A vous de déplacer correctement la raquette afin de la renvoyer le plus grand nombre de fois possibles. Pour chaque quille touchée, on pourra constater que le score, qui s'affiche en haut à droite de l'écran, se trouve bien augmenté de la valeur correspondant au nombre de points que vaut celle-ci.

Henri-Pierre Penel

```

820 REM *****
830 REM *
840 REM *      CONTROLE DU MOUVEMENT      *
850 REM *      DE LA RAQUETTE.            *
860 REM *
870 REM *****
880 LOCATE XR,23:PRINT CHR$(135);CHR$(136)
890 LET XRP=XR
900 LET XR=XR+INKEY(1)-INKEY(8)
910 IF XR<2 THEN LET XR=2
920 IF XR>20 THEN LET XR=20
930 IF XR<>XRP THEN LOCATE XRP,23:PRINT " "
940 LOCATE XR,23:PRINT CHR$(135);CHR$(136)
950 IF SND=0 THEN GOTO 1040
960 SOUND 1,(10*YB),0,0,1,0,0
970 SOUND 2,700-(YB*20),0,0,2,0,0
980 REM *****
990 REM *
1000 REM *      ANIMATION DU MOUVEMENT      *
1010 REM *      DE LA BALLE.                *
1020 REM *
1030 REM *****
1040 IF XB<3 OR XB>20 THEN LET MXB=-MXB
1050 LET XBP=XB:LET YBP=YB
1060 LET XB=XB+MXB
1070 LET YB=YB+MYB
1080 IF YB=2 THEN LET MYB=1
1090 LET RXT=INT(XB/2):LET RYT=YB-4
1100 REM *****
1110 REM *
1120 REM *      TEST QUILLE TOUCHEE.        *
1130 REM *
1140 REM *****
1150 IF RYT<1 OR RYT>4 OR XB>21 THEN GOTO 1260
1160 IF B(RXT,RYT)<>0 THEN GOTO 1260
1170 LET B(RXT,RYT)=1:LET MYB=-MYB
1180 IF XB<>2 AND XB<>21 THEN LET MXB=-MXB
1190 LOCATE (2*RXT),YB:PRINT " "
1200 REM *****
1210 REM *
1220 REM *      CALCUL DU SCORE.              *
1230 REM *
1240 REM *****
1250 LET SC=SC+(5*RYT):LOCATE 30,3:PRINT SC
1260 LOCATE XBP,YBP:PRINT " "
1270 LOCATE XB,YB:PRINT CHR$(137)
1280 IF YB<>22 THEN GOTO 1310
1290 IF XB=XR THEN LET MXB=-1:LET MYB=-1
1300 IF XB=XR+1 THEN LET MXB=1:LET MYB=-1
1310 IF YB<>23 THEN GOTO 1530
1320 REM *****
1330 REM *
1340 REM *      CALCUL DU CREDIT RESTANT.    *
1350 REM *
1360 REM *****
1370 LET CR=CR-1
1380 LOCATE XR,23:PRINT " "
1390 LOCATE XB,23:PRINT " "
1400 LOCATE 32,5:PRINT CR
1410 FOR I=1 TO 22
1420 SOUND 1,(10*I),0,0,1,0,0
1430 SOUND 2,500-(I*20),0,0,2,0,0
1440 FOR T=1 TO 60:NEXT T
1450 NEXT I
1460 IF CR=0 THEN GOTO 1590
1470 REM *****
1480 REM *
1490 REM *      REBOUCLAGE DU JEU.          *
1500 REM *
1510 REM *****
1520 GOTO 790
1530 GOTO 880
1540 REM *****
1550 REM *
1560 REM *      DEMANDE DE NOUVELLE PARITE.  *
1570 REM *
1580 REM *****
1590 LOCATE 25,15:PRINT "FIN DE JEU. "
1600 LOCATE 25,17:PRINT "POUR UNE AUTRE "
1610 LOCATE 25,19:PRINT "PARTIE TAPER"
1620 LOCATE 25,21:PRINT "LA TOUCHE 'P'."
1630 LET K$=UPPER$(INKEY$)
1640 IF K$<>"P" THEN GOTO 1630
1650 GOTO 10

```